Question: From the BodePlot function, I would like to plot the vertical line at horizontal axis equal to 10 as in the magnitude plot below and indicate the coordinate of the intersection point. 

Could anyone help me to do this?

Thank you.

```
BodePlot[1/(1 + s/10 + s^2/100)]

```

[![enter image description here](https://i.stack.imgur.com/Huijt.png)](https://i.stack.imgur.com/Huijt.png)

Answer: 
```
bplts = BodePlot[1/(1 + s/10 + s^2/100), 
 GridLines -> {{{10}, None}, {{10}, None}}, GridLinesStyle -> Red, 
 Mesh -> {{{10}}, {{10}}}, MeshStyle -> PointSize[Large]]

```

![Mathematica graphics](https://i.stack.imgur.com/YuTmd.png)

**Update:** You can show the coordinates of the intersection by post-processing the graphics output to add `Tooltip` or `Text`:

```
(Normal /@ bplts) /. Point[p__] :> Tooltip[Point[p], p]

```

![Mathematica graphics](https://i.stack.imgur.com/sVkIc.png)

```
(Normal /@ bplts) /. 
 Point[p__] :> {Point[p], Red, Text[Style[p, 16], p, Left]}

```

![Mathematica graphics](https://i.stack.imgur.com/lzwWd.png)
```
BodePlot[1/(1 + s/10 + s s/100),
 Epilog -> Line[{{1, -100}, {1, .11}}]]

```

or

```
BodePlot[1/(1 + s/10 + s s/100),
 Epilog -> {{Red, Line[{{1, -100}, {1, .11}}], 
            {PointSize[0.05], Point[{1, .11}]}}, 
            {Red, Line[{{1, -90}, {1, -200}}]}}]

```

[![enter image description here](https://i.stack.imgur.com/svP0S.png)](https://i.stack.imgur.com/svP0S.png)

If you want to have the points calculated automatically for your choice of abscissa:

```
mysvalue = 1;
f[s_] := 1/(1 + s/10 + s s/100);
BodePlot[f[s],
 Epilog -> {{Red, 
    Line[{{mysvalue, -100}, {mysvalue, f[mysvalue]}}], 
          {PointSize[0.05], 
    Point[{mysvalue, f[mysvalue]}]}}, 
   {Red, Line[{{mysvalue, -90}, {mysvalue, -200}}]}}]

```

Apparently abscissas are represented on a log scale here.